



3.0.0 **alpha build**
by Hiren H. Patel [hpatel6@ucsc.edu]

Primer

November 13, 2020

Contents

1	Installing and loading the package	2
2	Using the Wolfram Documentation Center	2
3	Preliminaries: Input	3
3.1	Entering Lorentz dot products (<i>e.g.</i> k.p)	3
3.2	Entering four-vectors (<i>e.g.</i> P_μ , γ_μ) and tensors (<i>e.g.</i> g_{μ,ν})	3
4	Tutorial 1: Taking traces	4
4.1	Entering special glyphs (<i>e.g.</i> 1 , γ , g)	4
4.2	Tree-level $Z^0 \rightarrow f\bar{f}$ decay width	5
5	Tutorial 2: Computing one-loop diagrams	8
5.1	QED vacuum polarization (Spur)	8
5.2	Standard model $H \rightarrow gg$ decay width (Spur)	11
5.3	Electron self-energy function (DiracMatrix)	14
5.4	Electron anomalous magnetic moment (Projector)	17
5.5	One-loop QCD correction to $\gamma^* \rightarrow q\bar{q}$ (FermionLine , u, v)	20
5.6	Tree-level $Z^0 \rightarrow f\bar{f}$ decay width (Discontinuity)	23
6	Tips for a pleasant computing experience	25
A	Space-time conventions	29

A message from the author

In this tutorial, I provide an introduction to the use of *Package-X*. This tutorial contains a series of simple examples to illustrate basic usage. To follow them, I expect that the reader has some familiarity in the evaluation of dimensionally regulated one-loop integrals, and in the use of *Mathematica*. But because I designed this package with a shallow learning curve in mind, they are (hopefully) easy to follow, and it should not take too long before becoming acquainted with the package.


1 Installing and loading the package

Package-X can be installed on any computer that runs *Mathematica* 10.1 or higher. The latest version of the program, including expansion packs, is hosted at the hepforge project page <http://packagex.hepforge.org>.

After downloading and uncompressing the zip file, place the folder `/X` and all its contents within the directory `$UserBaseDirectory/Applications/`. You can easily open `$UserBaseDirectory` on your system by running

```
In[1]:= SystemOpen[$UserBaseDirectory]
```

within a *Mathematica* session. After placing the folder `/X` into the `Applications` directory, it is necessary to restart *Mathematica* for the system to find and merge the *Package-X* help files with the *Mathematica Documentation Center*. For uninstallation, simply remove the folder from the directory; the help files will automatically be removed.

On an American keyboard, the grave accent ` is found just below the  key.

You can load the package in the current session with the command:


```
In[1]:= << X`
```

```
Package-X v3.0.0, by Hiren H. Patel  
For more information, see the guide
```

This will activate all definitions made by *Package-X*, and the functions defined within are ready for use.


2 Using the Wolfram Documentation Center

Included with *Package-X* is a set of documentation files which is automatically merged with the Wolfram Documentation Center upon starting *Mathematica*. The documentation files contain details and examples for every symbol/function defined by *Package-X*. One way to view these files is by visiting the guide page via the printed link when *Package-X* is initialized.

Alternatively, pressing  when the cursor is within a function name will bring up the relevant help page for that function.

Finally, brief information about a function can be obtained by using `?` followed by its name. For example,

```
In[1]:= ? LoopRe fine
```



And additional information can be obtained by clicking on the info icon () in

Tensor	Standard- Form	FullForm
p_μ	\mathbf{p}_μ	<code>LTensor[p, μ]</code>
γ_μ	$\boldsymbol{\gamma}_\mu$	<code>LTensor[DiracG, μ]</code>
$e_\mu(p, m)$	$\mathbf{e}[\mathbf{p}, m]_\mu$	<code>LTensor[PolVecE[p, m], μ]</code>
$g_{\mu\nu}$	$\mathbf{g}_{\mu, \nu}$	<code>LTensor[MetricG, μ, ν]</code>
$\varepsilon_{\mu\nu\rho\varsigma}$	$\boldsymbol{\varepsilon}_{\mu, \nu, \rho, \varsigma}$	<code>LTensor[LeviCivitaE, $\mu, \nu, \rho, \varsigma$]</code>

Table 1: Some *Package-X* tensors in `StandardForm` (suitable for notebook input) and `FullForm` (suitable for command-line input)

the upper-right corner which takes you to the help page in the Documentation Center.

3 Preliminaries: Input

Package-X is designed with increased readability of input code by having its syntax closely resemble corresponding expressions that would be written by hand. Therefore, it is natural that special input methods and symbols would be an integral part of the input for functions defined by *Package-X*. In this section, the preliminaries are introduced.

3.1 Entering Lorentz dot products (e.g. $\mathbf{k} \cdot \mathbf{p}$)

Once *Package-X* is loaded, the shortcut infix operator ‘.’ is remapped to a new function `LDot` that represents the Lorentz dot product.

```
In[1]:= (*Lorentz dot product in Package-X*)
         $\mathbf{p} \cdot \mathbf{k}$ 
Out[1]= k . p
```

The built-in *Mathematica* function `Dot` is still available to you to perform standard matrix multiplication or inner products, but it can no longer be invoked by the shortcut.

```
In[2]:= (*Mathematica's inner product*)
        Dot[{a, b, c}, {d, e, f}]
Out[2]= a d + b e + c f
```

You can input dotted four-vectors in any order; *Mathematica* will canonically order them freely as in the example above. To save input time, you can enter dot products of vector sums or differences, e.g. $(\mathbf{p} + \mathbf{k}) \cdot (\mathbf{q} - \mathbf{k})$.

3.2 Entering four-vectors (e.g. \mathbf{p}_μ , $\boldsymbol{\gamma}_\mu$) and tensors (e.g. $\mathbf{g}_{\mu, \nu}$)

Computations performed with *Package-X* are intended to be fully Lorentz covariant. Thus the distinction between covariant p^μ and contravariant p_μ indices is unnecessary. All indices will appear in “contravariant” position in your notebook (e.g. \mathbf{p}_μ). You can enter four-vectors and higher rank tensors (see Table 1) by placing their indices in subscript position (short-cut `ctrl`+`-`), and *Package-X* will automatically parse the syntax as `LTensor`.

The function `Contract` is defined to recognize repeated indices and will form the appropriate dot-products.

Glyph	Alias	Description
1	<code>esc</code> 11 <code>esc</code>	unit Dirac matrix
γ	<code>esc</code> g <code>esc</code>	Dirac matrix
σ	<code>esc</code> s <code>esc</code>	Dirac sigma matrix
γ5	<code>esc</code> g <code>esc</code> 5	Gamma-5 matrix
PL	<code>esc</code> PL <code>esc</code>	left chiral projector
PR	<code>esc</code> PR <code>esc</code>	right chiral projector
g_{□,□}	<code>esc</code> gg <code>esc</code>	metric tensor
ε_{□,□,□,□}	<code>esc</code> levi <code>esc</code>	Levi-Civita symbol

Table 2: Aliases for notebook entry of commonly used symbols

```
In[1]:= Contract [pμ kν kμ]
Out[1]= k . p kν
```

Note that subscripts tied to sums/differences of symbols are understood to be four-vector sums/differences (*e.g.* $(\mathbf{p} + 2\mathbf{k})_\mu \equiv (p_\mu + 2k_\mu)$), and can save you time during input.

Pay particular attention to the syntax for higher rank tensors, such as $\mathbf{g}_{\mu,\nu}$ and $\epsilon_{\mu,\nu,\rho,\varsigma}$ (see next section), for which the indices are delimited by commas. Forgetting to include the comma separators would be interpreted by *Mathematica* as multiplication of the corresponding symbols inside the subscript.

4 Tutorial 1: Taking traces

4.1 Entering special glyphs (e.g. **1**, **γ**, **g**)

Consult the Package-X guide page in the Wolfram Documentation Center for a complete list.

Let us begin using *Package-X* to evaluate traces over products of Dirac matrices with the function **Spur**. In order to maintain readability of code, certain glyphs have been defined for input. Special aliases (keystroke combination `esc`, ..., `esc`) have been reserved for rapid notebook entry. Some commonly used symbols are in table 2.

We begin by evaluating the following trace:

$$\text{Tr}[(\not{p}_1 - m) \gamma_\mu (\not{p}_2 - m) \gamma_\nu]$$

The corresponding input is as follows (consult table 2 for entering **γ** and **1**, and section 3.2 for entering subscripted indices)¹:

```
In[1]:= Spur[γ.p1 - m1, γμ, γ.p2 - m1, γν]
Out[1]= 4 p1ν p2μ + 4 p1μ p2ν + 4 m2 gμ, ν - 4 p1 . p2 gμ, ν
```

Note that you must enter the Feynman slash \not{p} by typing out the scalar dot-product of γ and p in full. Also, the glyph **1** stands for the unit 4×4 Dirac matrix which, although conventionally implicit in written form, must be supplied explicitly for robustness of the syntax. It is to be used multiplicatively when a constant is present (in this example **m**) that does not multiply any other object in spinor space (such as **γ**, **σ**, or **γ5**).

As a general rule, you can add objects in spinor-space together *e.g.*

$$\text{Tr}[1 + c\gamma_5] \equiv \text{Spur}[1 + c\gamma_5],$$

¹Greek symbols can also be inserted rapidly using *Mathematica*'s pre-defined aliases. For example, Greek letter μ is obtainable from `esc` m `esc`; symbol ν from `esc` n `esc`, *etc.* Although Greek symbols are not required for entering subscripts, they make for readable code.

but you must set off products by comma delimiters to preserve ordering of matrix-multiplication *e.g.*

$$\text{Tr}[\gamma_\mu \gamma_\nu \gamma_\lambda \gamma^\mu] \equiv \mathbf{Spur}[\gamma_\mu, \gamma_\nu, \gamma_\lambda, \gamma_\mu].$$

4.2 Tree-level $Z^0 \rightarrow f\bar{f}$ decay width

In this next example, we will compute the partial width of the Z^0 gauge boson disintegration into two fermions f . Remember that *Package-X* is *not* FEYN-CALC: it will not give a partial width starting from the Feynman rules. Instead, **Spur** can be used to evaluate the traces that appear in the intermediate steps of the computation. Using the Feynman rules for electroweak theory, we find that the tree-level amplitude is given by:

$$i\mathcal{M} = \bar{u}_{s'}(\mathbf{p}') \frac{-ie\gamma^\mu}{\sin 2\theta_W} (g_V^f - g_A^f \gamma_5) v_s(\mathbf{p}) \epsilon_\mu^{[\lambda]}(\mathbf{k})$$

To obtain the unpolarized decay width, square the amplitude, average over initial helicity states $\frac{1}{3} \sum_\lambda$, and sum over final spins and colors $\sum_{s,s'} \sum_{\text{color}}$.

$$|\overline{\mathcal{M}}|^2 = \frac{e^2}{\sin^2(2\theta_W)} N_c \sum_{s,s'} \left(u_{s'}(\mathbf{p}') \bar{u}_{s'}(\mathbf{p}') \gamma^\mu (g_V^f - g_A^f \gamma_5) \right. \\ \left. \times v_s(\mathbf{p}) \bar{v}_s(\mathbf{p}) (g_V^f + g_A^f \gamma_5) \gamma^\nu \right) \frac{1}{3} \sum_\lambda \epsilon_\mu^{[\lambda]}(\mathbf{k}) \epsilon_\nu^{*[\lambda]}(\mathbf{k}) \quad (1)$$

$$= \frac{e^2}{\sin^2(2\theta_W)} \frac{N_c}{3} \text{Tr} \left[(\not{p}' + m_f) \gamma^\mu (g_V^f - g_A^f \gamma_5) \right. \\ \left. \times (\not{p} - m_f) (g_V^f + g_A^f \gamma_5) \gamma^\nu \right] \left(-g_{\mu\nu} + \frac{k_\mu k_\nu}{m_Z^2} \right) \quad (2)$$

Now, use *Package-X* to compute the trace appearing in equation 2 (we represent $p = \mathbf{p1}$ and $p' = \mathbf{p2}$):

```
In[1]:= Spur[γ. p2 + mf 1, γμ, gV 1 - gA γ5, γ. p1 - mf 1, gV 1 + gA γ5, γν]
```

```
Out[1]= 4 gA^2 p1ν p2μ + 4 gV^2 p1ν p2μ + 4 gA^2 p1μ p2ν + 4 gV^2 p1μ p2ν + 4 gA^2 mf^2 gμν -
4 gV^2 mf^2 gμν - 4 gA^2 p1. p2 gμν - 4 gV^2 p1. p2 gμν + 8 i gA gV εμνρ{p1},{p2}
```

This is the SCHOONSCHIP notation, introduced by Veltman The symbol $\epsilon_{\mu,\nu,\{p1\},\{p2\}} = \epsilon^{\mu\nu\alpha\beta} p1_\alpha p2_\beta$ in the output represents the double contraction of the Levi-Civita antisymmetric symbol.

Then, we can contract the result with the Z^0 polarization sum $(-g_{\mu\nu} + k_\mu k_\nu / m_Z^2)$

```
In[2]:= Contract[Spur[γ. p2 + mf 1, γμ, gV 1 - gA γ5, γ. p1 - mf 1, gV 1 + gA γ5, γν]
(-gμν + kμ kν / mZ^2)]
```

```
Out[2]= -4 d gA^2 mf^2 + 4 d gV^2 mf^2 - 8 gA^2 p1. p2 + 4 d gA^2 p1. p2 -
8 gV^2 p1. p2 + 4 d gV^2 p1. p2 + 1/mZ^2 (4 gA^2 mf^2 k.k - 4 gV^2 mf^2 k.k +
8 gA^2 k. p1 k. p2 + 8 gV^2 k. p1 k. p2 - 4 gA^2 k.k p1. p2 - 4 gV^2 k.k p1. p2)
```

The output contains a number of dot-products of the external momenta. We can give the on shell conditions for the various 4-vectors by first storing them as a list of **Rules** (infix shortcut \rightarrow which gets auto-replaced with \rightarrow)

```
In[3]:= onShell = {k.k → mZ2, p1.p1 → mf2, p2.p2 → mf2,
                p1.p2 → (mZ2 - 2 mf2)/2, k.p1 → mZ2/2, k.p2 → mZ2/2};
```

and then applying them by using **ReplaceAll** (suffix shortcut /.).

Finally, the symbol d (alias: `esc` dim `esc`) in the output represents the number of space-time dimensions. For purposes of this tree-level computation we would like to set $d = 4$. In *Package-X*, the recommended way to achieve this is to apply **LoopRefine** on the output. In the next section, we will use **LoopRefine** frequently to compute one-loop integrals.

So our program to compute the $Z \rightarrow f\bar{f}$ matrix element (square) looks like this:

```
In[3]:= onShell = {k.k → mZ2, p1.p1 → mf2, p2.p2 → mf2, p1.p2 → (mZ2 - 2 mf2)/2,
                k.p1 → mZ2/2, k.p2 → mZ2/2};

In[4]:= Contract [Spur [γ.p2 + mf 1, γμ, gV 1 - gA γ5, γ.p1 - mf 1, gV 1 + gA γ5, γν]
                (-gμ,ν + kμ kν / mZ2)] /. onShell // LoopRefine

Out[4]= -4 (4 gA2 mf2 - 2 gV2 mf2 - gA2 mZ2 - gV2 mZ2)
```

There are two other (but more precarious) ways to set $d = 4$:

- Appending a replacement at the end of the line `/.d → 4`. Be warned that making this replacement at an intermediate stage of computing one-loop integrals can lead to incorrect finite results.
- Declare $d = 4$; . This method is especially dangerous, and should be used only if tree-level computations will be carried out in the current *Mathematica* session. Be sure to **Clear**[d] before using **LoopIntegrate** and **LoopRefine** to compute one-loop integrals!

We can now manipulate this result using *Mathematica*'s built-in algebraic manipulation routines. For example, we can reorganize the output using **Collect**:

```
In[5]:= Collect [% , {mf, mZ}]

Out[5]= 4 (-4 gA2 + 2 gV2) mf2 + 4 (gA2 + gV2) mZ2
```

Therefore, the spin-averaged matrix element is

$$|\overline{\mathcal{M}}|^2 = \frac{e^2}{\sin^2(2\theta_W)} \frac{N_c}{3} 4 \left[((g_V^f)^2 + (g_A^f)^2) m_Z^2 + 2((g_V^f)^2 - 2(g_A^f)^2) m_f^2 \right] \quad (3)$$

from which we can finally deduce the partial width:

$$\Gamma = \frac{\alpha m_Z}{\sin^2(2\theta_W)} \left(\frac{N_c}{3}\right) \sqrt{1 - \frac{4m_f^2}{m_Z^2} \left[((g_V^f)^2 + (g_A^f)^2) + 2((g_V^f)^2 - 2(g_A^f)^2) \frac{m_f^2}{m_Z^2} \right]}. \quad (4)$$

In section 5.6, I will show you how to arrive at the same result using unitarity of the S -matrix.

5 Tutorial 2: Computing one-loop diagrams

In this section, I provide examples to compute one-loop integrals. The basic procedure of computing the integrals using *Package-X* consists of three steps:

1. Run **LoopIntegrate** to express the integral in terms of scalar coefficient functions. (If needed, projections onto form factors should be made at this stage)
2. Supply on shell and other kinematic conditions by calling **Replace**[*expr*, *rules*], (or equivalently *expr* /. {*rules*}). **Never manually make the $d \rightarrow 4$ replacement at this stage.**
3. Run **LoopRe fine** to safely take the $d \rightarrow 4$ limit, and to obtain the integral in terms of analytic expressions.

5.1 QED vacuum polarization (Spur)




Figure 1: One-loop vacuum polarization in QED

Calling **?LoopIntegrate** will display its syntax information.

We will begin with the concrete example of computing the one-loop QED vacuum polarization function $\Pi^{\mu\nu}(q)$. The Feynman diagram along with the momentum routing is shown in Fig. 1. The usage information for **LoopIntegrate** is summarized below:

$$\mathbf{LoopIntegrate}[num, k, \{\{q_0, m_0\}, \{q_1, m_1\}, \dots\}] = \left(\frac{i e^{-\gamma_E \epsilon}}{(4\pi)^{d/2}} \right)^{-1} \mu^{2\epsilon} \int \frac{d^d k}{(2\pi)^d} \frac{num}{(q_0^2 - m_0^2 + i\epsilon)(q_1^2 - m_1^2 + i\epsilon) \dots}$$

Here are some useful points concerning the syntax of **LoopIntegrate**:

- The first argument specifies the numerator, and should be a tensor (k_μ) or scalar-product (e.g. $k.p$ or k^2) polynomial in the integration variable.
- The second argument specifies the integration variable k .
- The third argument specifies the product of denominator factors, enclosed in a **List**. Every first element of the list q_i is a linear combination of the integration variable k and external momenta p , and every second element m_i is a mass variable.

It is important to keep in mind how **LoopIntegrate** normalizes the integrals:

An overall factor $i e^{-\gamma_E \epsilon} / (4\pi)^{d/2}$ has been removed from the output of **LoopIntegrate** (and hence also from **LoopRe fine**) to maintain readability of the final result. This means that an overall factor of $\frac{i}{16\pi^2}$ is omitted, and the constant $-\gamma_E + \ln(4\pi)$ that accompanies the $1/\epsilon$ poles in UV divergent and mass singular integrals is also missing. *Remember to restore these constants at the end of your computation.*

Start by applying the QED Feynman rules to write down the integral representation of the vacuum polarization function:

$$\begin{aligned}
i\Pi^{\mu\nu}(q) &= (-)\mu^{2\epsilon} \int \frac{d^d k}{(2\pi)^d} \text{Tr} \left[i e \gamma^\nu \frac{i(\not{k} + m)}{k^2 - m^2} i e \gamma^\mu \frac{i(\not{k} - \not{q} + m)}{(k - q)^2 - m^2} \right] \\
&= -e^2 \mu^{2\epsilon} \int \frac{d^d k}{(2\pi)^d} \frac{\text{Tr} [\gamma^\nu (\not{k} + m) \gamma^\mu (\not{k} - \not{q} + m)]}{(k^2 - m^2)((k - q)^2 - m^2)}
\end{aligned} \tag{5}$$

In the second line, the integral has been brought to a form that is ready for input.

```

In[1]:= vacuumPol = LoopIntegrate[Spur[γν, γ.k + m1, γμ, γ.(k - q) + m1], k,
      {{k, m}, {k - q, m}}]

Out[1]= qμ qν (8 PVB[0, 1, q.q, m, m] + 8 PVB[0, 2, q.q, m, m]) +
      gμ,ν (-4 PVA[0, m] + 2 q.q PVB[0, 0, q.q, m, m] + 8 PVB[1, 0, q.q, m, m])

```

Note that the numerator of the integrand is a Dirac trace, so the function **Spur** is used in the first argument of **LoopIntegrate**.

The output is a decomposition of the loop integral in terms of Passarino-Veltman tensor coefficient functions **PVA** and **PVB**. These functions cannot be numerically evaluated, but will be replaced with analytic expressions upon calling **LoopRefine**.

The next step is to supply kinematic conditions. But, since the polarization function is an off shell quantity, there is no information to give; so we'll skip this step. Call **LoopRefine** to obtain the vacuum polarization function:

```

In[2]:= LoopRefine[vacuumPol]

Out[2]= ( - 4 DiscB[q.q, m, m] (2 m^2 + q.q) - 4 (12 m^2 + 5 q.q) - 4 (1/ε + Log[μ^2/m^2]) ) qμ qν +
      ( 4/3 DiscB[q.q, m, m] (2 m^2 + q.q) + 4/9 (12 m^2 + 5 q.q) + 4/3 q.q (1/ε + Log[μ^2/m^2]) ) gμ,ν

```

LoopRefine always explicitly displays logarithmic UV and/or IR divergences as $1/\epsilon$ poles. If its output is free of ϵ , it is free of these divergences.

Consult the *Package-X guide page* for more information about **DiscB**, and other special functions.

A few notes concerning this output:

- In *Package-X*, UV and IR divergences of loop integrals are regulated by dimensional regularization with $d = 4 - 2\epsilon$. At one-loop, these divergences will appear as $1/\epsilon$ and $1/\epsilon^2$ poles near 4 dimensions, and will be accompanied by logarithms of the 't Hooft parameter μ (alias: `esc` mm `esc`). In this example, the UV divergence is clearly visible.
- The function **DiscB** is an abbreviation for a more complicated expression that contains the normal threshold discontinuity of the Passarino-Veltman B_0 function. You can call **DiscExpand** to display it in terms of elementary functions. Special simple cases of **DiscB** will be substituted automatically.

- A rank two-tensor can be written as a sum of parts transverse and longitudinal to a given vector v :

$$\Pi^{\mu\nu} = \left(g^{\mu\nu} - \frac{v^\mu v^\nu}{v^2}\right)\Pi_T + \left(\frac{v^\mu v^\nu}{v^2}\right)\Pi_L \quad (6)$$

Applying **Transverse** or **Longitudinal** to a rank-2 tensor will give the corresponding projection. The projection should be made *before* applying any on shell conditions or calling **LoopRe fine**.

- Expressions generated by **LoopRe fine** are valid for all real external invariants and positive internal masses. When numerically evaluated, the $+i\epsilon$ prescription is appropriately taken into account and the correct real and imaginary parts are returned.

With these points in mind, let us extract the longitudinal and transverse parts of the vacuum polarization tensor:

```
In[3]:= vacuumPol // Longitudinal // LoopRe fine
        vacuumPol // Transverse // LoopRe fine // DiscExpand

Out[3]= 0

Out[4]= 4/9 (12 m^2 + 5 q . q) + 4/3 q . q (1/epsilon + Log[mu^2/m^2]) +
        4 sqrt[q . q (-4 m^2 + q . q)] (2 m^2 + q . q) Log[ (2 m^2 - q . q + sqrt[q . q (-4 m^2 + q . q)]) / (2 m^2) ] / (3 q . q)
```

As expected based on the QED Ward identities, the vacuum polarization function has no longitudinal component. We can write down the (bare) vacuum polarization at one-loop:

$$i\Pi^{\mu\nu} = -e^2 \frac{i}{16\pi^2} \left[-\text{computer output} \right] \quad (7)$$

$$\begin{aligned} &= \frac{-ie^2}{16\pi^2} \left[\frac{4}{3} q^2 \left(\frac{1}{\epsilon} - \gamma_E + \ln(4\pi) + \ln\left(\frac{\mu^2}{m^2}\right) \right) + \frac{4}{9} (12m^2 + 5q^2) \right. \\ &\quad \left. + \frac{4}{3q^2} \sqrt{q^2(q^2 - 4m^2)} (2m^2 + q^2) \ln\left(\frac{2m^2 - q^2 + \sqrt{q^2(q^2 - 4m^2)}}{2m^2}\right) \right] \\ &\quad \times \left(g^{\mu\nu} - \frac{q^\mu q^\nu}{q^2} \right) \end{aligned} \quad (8)$$

5.2 Standard model $H \rightarrow gg$ decay width (**Spur**)

In this example, we will compute the partial width of the Higgs boson to two gluons mediated by a top quark loop. The Feynman diagrams along with the momentum routing is shown in Fig. 2. We shall perform the loop integral without anticipating the structure of the final answer based on Ward identities.




Figure 2: Momentum routing for the one-loop calculation of $H \rightarrow gg$ amplitude

Applying the Feynman rules of electroweak interactions, the amplitude is:

$$\begin{aligned}
 i\mathcal{M} &= \mu^{2\epsilon} \int \frac{d^d k}{(2\pi)^d} (-1) \text{Tr}_{\text{color}} \left[\left(\frac{-iy_t}{\sqrt{2}} \right) \frac{i(\not{k} - \not{q} + m_t)}{(k-q)^2 - m_t^2} (-igT^b \gamma^\nu) \right. \\
 &\quad \times \frac{i(\not{k} - \not{p} + m_t)}{(k-p)^2 - m_t^2} (-igT^a \gamma^\mu) \left. \frac{i(\not{k} + m_t)}{k^2 - m_t^2} \right] \epsilon_\mu^*(\mathbf{p}) \epsilon_\nu^*(\mathbf{p}') + \left(\begin{smallmatrix} p \leftrightarrow p' \\ \mu \leftrightarrow \nu \\ a \leftrightarrow b \end{smallmatrix} \right) \\
 &= \frac{-y_t g^2}{2\sqrt{2}} \delta_{\text{color}}^{ab} \epsilon_\mu^*(\mathbf{p}) \epsilon_\nu^*(\mathbf{p}') \\
 &\quad \times \left(\mu^{2\epsilon} \int \frac{d^d k}{(2\pi)^d} \frac{\text{Tr}[(\not{k} - \not{q} + m_t) \gamma^\nu (\not{k} - \not{p} + m_t) \gamma^\mu (\not{k} + m_t)]}{[(k-q)^2 - m_t^2][(k-p)^2 - m_t^2][k^2 - m_t^2]} + \left(\begin{smallmatrix} p \leftrightarrow p' \\ \mu \leftrightarrow \nu \end{smallmatrix} \right) \right)
 \end{aligned}$$

Now we use *Package-X* to carry out the integral in round parenthesis. In the code we represent $p = \mathbf{p1}$, $p' = \mathbf{p2}$ and $m_t = \mathbf{m}$:

```

In[1]:= LoopIntegrate[Spur[γ.(k-q)+m1, γν, γ.(k-p1)+m1, γμ, γ.k+m1],
  k, {{k-q,m},{k-p1,m},{k,m}}]+
LoopIntegrate[Spur[γ.(k-q)+m1, γμ, γ.(k-p2)+m1, γν, γ.k+m1],
  k, {{k-q,m},{k-p2,m},{k,m}}]

Out[1]= qμqν(8mPVC[0,0,1,p1.p1,p1.p1-2p1.q+q.q,q.q,m,m,m]+
  16mPVC[0,0,2,p1.p1,p1.p1-2p1.q+q.q,q.q,m,m,m])+<<9>>

```

This results in an expression some 20 lines long involving **PVB** and **PVC**. If we apply **LoopRe fine** at this stage, we would arrive at a *very complicated* analytic representation of the completely off-shell Hgg Green function.

But, by supplying the kinematic conditions relevant for the on shell decay process, we will find a much simpler analytic expression. The kinematic relations for this process are:

*Resist the temptation to denote the Higgs mass with a subscripted symbol $\mathbf{m_H}$, as *Package-X* would erroneously interpret it as a four-vector m with index H .*

$$\begin{array}{lll}
 p^2 = 0 & \mathbf{p1.p1} \rightarrow 0 & q^\mu = (p+p')^\mu \quad \mathbf{q} \rightarrow \mathbf{p1} + \mathbf{p2} \\
 p'^2 = 0 & \mathbf{p2.p2} \rightarrow 0 & p \cdot q = \frac{1}{2} m_H^2 \quad \mathbf{p1.q} \rightarrow \mathbf{mH^2/2} \\
 q^2 = m_H^2 & \mathbf{q.q} \rightarrow \mathbf{mH^2} & p' \cdot q = \frac{1}{2} m_H^2 \quad \mathbf{p2.q} \rightarrow \mathbf{mH^2/2}
 \end{array}$$

There are still two more important conditions we should supply. Remember that external on shell (massive or massless) gauge boson polarization vectors satisfy

the transversality condition $p.\epsilon(p) = 0$. For the case at hand, we can impose transversality of the gluon polarization vectors with the following replacements:

$$\begin{aligned} p.\epsilon(\mathbf{p}) = 0 &\Rightarrow \mathbf{p1}_\mu \rightarrow 0 \\ p'.\epsilon(\mathbf{p}') = 0 &\Rightarrow \mathbf{p2}_\nu \rightarrow 0 \end{aligned}$$

After including these eight conditions (which we do in the same input cell), we get a much shorter expression:

```
In[2]:= LoopIntegrate[Spur[γ.(k-q)+m1, γν, γ.(k-p1)+m1, γμ, γ.k+m1],
  k, {{k-q, m}, {k-p1, m}, {k, m}}] +
LoopIntegrate[Spur[γ.(k-q)+m1, γμ, γ.(k-p2)+m1, γν, γ.k+m1],
  k, {{k-q, m}, {k-p2, m}, {k, m}}]/.
{p1.p1→0, p2.p2→0, q.q→mH2, q→p1+p2, p1.q→mH2/2, p2.q→mH2/2}/.
{p1μ→0, p2ν→0}

Out[2]= 2 p1νp2μ (8 mPVC[0, 0, 1, 0, 0, mH2, m, m, m] + 16 mPVC[0, 0, 2, 0, 0, mH2, m, m, m]) +
2 p1νp2μ (4 mPVC[0, 0, 0, 0, 0, mH2, m, m, m] +
8 mPVC[0, 0, 1, 0, 0, mH2, m, m, m] + 16 mPVC[0, 1, 1, 0, 0, mH2, m, m, m]) +
2 gμ,ν (-4 mPVB[0, 0, mH2, m, m] -
2 m mH2 PVC[0, 0, 0, 0, 0, mH2, m, m, m] + 16 mPVC[1, 0, 0, 0, 0, mH2, m, m, m])
```

The final step of the computation is to call **LoopRefine** (again in the same input cell). Applying **Simplify** should collect and factor out an overall tensor structure that is consistent with the Ward identity on both external gluons:

```
In[3]:= LoopIntegrate[Spur[(k-q).γ+m1, γν, (k-p1).γ+m1, γμ, k.γ+m1],
  k, {{k-q, m}, {k-p1, m}, {k, m}}] +
LoopIntegrate[Spur[(k-q).γ+m1, γμ, (k-p2).γ+m1, γν, k.γ+m1],
  k, {{k-q, m}, {k-p2, m}, {k, m}}]/.
{p1.p1→0, p2.p2→0, q.q→mH2, q→p1+p2, p1.q→mH2/2, p2.q→mH2/2}/.
{p1μ→0, p2ν→0} // LoopRefine // Simplify

Out[3]= 
$$\frac{2 m \left( 4 m H^2 + (4 m^2 - m H^2) \operatorname{Log}\left[\frac{2 m^2 - m H^2 + \sqrt{-4 m^2 m H^2 + m H^4}}{2 m^2}\right]^2 \right) \left( -2 p1_\nu p2_\mu + m H^2 g_{\mu,\nu} \right)}{m H^4}$$

```

Note that a finite result is obtained, independent of the dimensional regulator ϵ and scale μ . With this we may proceed to extract the partial width.

We can parametrize the amplitude as follows:

$$i\mathcal{M} = (g^{\mu\nu} - \frac{2p'^\mu p^\nu}{m_H^2}) \epsilon_\mu^*(\mathbf{p}) \epsilon_\nu^*(\mathbf{p}') \delta_{\text{color}}^{ab} F_{1/2}(m_H, m_t). \quad (9)$$

Here $F_{1/2}(m_H, m_t)$ is the dimensionless loop function, which we can read off the

Mathematica output (remembering the implicit $i/(16\pi^2)$):

$$F_{1/2}(m_H, m_t) = \frac{i}{16\pi^2} \frac{-y_t g^2}{2\sqrt{2}} \times \frac{2m_t}{m_H^2} \left[4m_H^2 + (4m_t^2 - m_H^2) \ln^2 \left(\frac{2m_t^2 - m_H^2 + \sqrt{-4m_t^2 m_H^2 + m_H^4}}{2m_t^2} \right) \right]$$

Note that the expression is valid in both domains $m_H > 2m_t$ and $m_H < 2m_t$. As mentioned earlier, the $+i\epsilon$ prescription is appropriately taken into account by **LoopRe fine**. I invite you to make a plot of the real and imaginary parts of the loop function by running the following two lines:

```
In[4]:= f[mH_, m_] = Coefficient[%, gμ, ν]
In[5]:= Plot[{Re[f[mH, 1]], Im[f[mH, 1]]}, {mH, 0, 5}]
```

After squaring the amplitude, summing over the final state gluon spins and colors, and integrating over the gluon momenta, we obtain the partial width:

$$\Gamma_{H \rightarrow gg} = \frac{1}{2\pi m_H} |F_{1/2}(m_H, m_t)|^2. \quad (10)$$

5.3 Electron self-energy function (**DiracMatrix**)

In the next three examples, I will show you how to deal with loop integrals with open fermion lines. Such integrals are analytically cumbersome to work with, and so, I have designed *Package-X* to provide a number of ways to approach such integrals:

- Use *Package-X*'s **FermionLine** (for on shell external fermions) or **DiracMatrix** (for off shell external fermions) objects to compute the integral.




Figure 3: One-loop electron self-energy

We will start with the simpler example of calculating the electron self-energy function in quantum electrodynamics. The Feynman diagram is shown in Figure 3. For this particular example, we will also work in the general class of covariant gauges, where the photon propagator takes the form

$$-i\tilde{D}^{\mu\nu}(k) = \frac{-i}{k^2 + i\epsilon} \left[g^{\mu\nu} - (1 - \xi) \frac{k^\mu k^\nu}{k^2 + i\epsilon} \right] = \frac{-i}{(k^2 + i\epsilon)^2} \left[g^{\mu\nu} k^2 - (1 - \xi) k^\mu k^\nu \right].$$

In the second form, the denominator factors have been brought together, so that the one-loop integral representation of the self-energy function can be written more compactly:

$$-i\Sigma(p) = -e^2 \mu^{2\epsilon} \int \frac{d^d k}{(2\pi)^d} \frac{\gamma_\nu (\not{p} - \not{k} + m) \gamma_\mu (g^{\mu\nu} k^2 - (1 - \xi) k^\mu k^\nu)}{[(p - k)^2 - m^2](k^2)^2}. \quad (11)$$

We will start by working on the numerator, which is a matrix function in spinor space. To enter a product of Dirac matrices without computing its trace, use **DiracMatrix**:

```
In[1]:= num = DiracMatrix[γν, γ. (p - k) + m 1, γμ] (gμν k.k - (1 - ξ) kμ kν) // Contract

Out[1]= -(1 - ξ) DiracMatrix[γ.k, 1 m - γ.k + γ.p, γ.k] +
         DiracMatrix[γν, 1 m - γ.k + γ.p, γν] k.k
```

Notice how the syntax of **DiracMatrix** is identical to that of **Spur**. We have also applied **Contract** to contract the tensors into the Dirac matrices. As an aside, we can expand the products of Dirac matrices to bring products such as **k.k** together using **FermionLineReduce**:

```
In[2]:= num // FermionLineReduce

Out[2]= DiracMatrix[γ.p] (2 k.k - d k.k + (1 - ξ) k.k) +
         DiracMatrix[] (m d k.k + m (-1 + ξ) k.k) +
         DiracMatrix[γ.k]
         (-2 k.k + d k.k + (1 - ξ) k.k + 2 (-1 + ξ) k.p)
```

TraditionalForm[expr] The object **DiracMatrix[]** with no arguments stands for the unit Dirac matrix. *Package-X* displays 'expr' in more compact mathematical notation.

Package-X Primer 14

trix. Note that applying **FermionLineReduce** is not really necessary since **LoopIntegrate** automatically uses an internal version of it to perform the numerator algebra.

Now call **LoopIntegrate** to perform the integral. Notice that in (11) the photon denominator factor k^2 is repeated, *i.e.* raised to the second power. To input this repetition to **LoopIntegrate**, we can supply a third element to the corresponding denominator specifier **{k, 0, 2}** to indicate the power of that denominator factor:

```
In[3]:= int = LoopIntegrate[num, k, {{k - p, m}, {k, 0, 2}}]

Out[3]= DiracMatrix[]
        (-mPVB[0, 0, p.p, 0, m] + m dPVB[0, 0, p.p, 0, m] + m ξ PVB[0, 0, p.p, 0, m]) +
        DiracMatrix[γ.p] (3 PVB[0, 0, p.p, 0, m] - dPVB[0, 0, p.p, 0, m] -
        ξ PVB[0, 0, p.p, 0, m] + 2 PVB[0, 1, p.p, 0, m] -
        dPVB[0, 1, p.p, 0, m] - m2 PVB[0, 1, p.p, 0, m, Weights → {2, 1}] +
        m2 ξ PVB[0, 1, p.p, 0, m, Weights → {2, 1}] +
        p.p PVB[0, 1, p.p, 0, m, Weights → {2, 1}] -
        ξ p.p PVB[0, 1, p.p, 0, m, Weights → {2, 1}])
```

Note the appearance of *weighted* Passarino-Veltman **PVB** functions in the output, with the final argument indicating the weight vector (2, 1). Such exotic Passarino-Veltman functions will occur when integrals with repeated propagators are computed. In general, they have better UV behavior (but worse IR behavior) than the corresponding unweighted Passarino-Veltman functions.

Since the self-energy function is an off shell quantity, there are no kinematic relations to include; so now we apply **LoopRe fine**:

```
In[5]:= LoopRe fine[int]

Out[5]= DiracMatrix[]
        (2 m (2 + ξ) + m (3 + ξ) (1/ε + Log[μ2/m2]) + m (3 + ξ) (-m2 + p.p) Log[m2/(m2 - p.p)]) +
        DiracMatrix[p.γ] (-ξ (m2 + p.p)/p.p - ξ (1/ε + Log[μ2/m2]) - ξ (-m4 + (p.p)2) Log[m2/(m2 - p.p)]/(p.p)2)
```

We can write down the final answer:

$$\begin{aligned}
-i\Sigma(p) = & \\
& p \frac{-ie^2}{16\pi^2} \left[-\xi \left(\frac{1}{\epsilon} - \gamma_E + \ln(4\pi) + \ln\left(\frac{\mu^2}{m^2}\right) \right) - \frac{\xi(m^2 + p^2)}{p^2} - \frac{\xi(-m^4 + (p^2)^2) \ln\left(\frac{m^2}{m^2 - p^2}\right)}{(p^2)^2} \right] \\
& + m \frac{-ie^2}{16\pi^2} \left[(3 + \xi) \left(\frac{1}{\epsilon} - \gamma_E + \ln(4\pi) + \ln\left(\frac{\mu^2}{m^2}\right) \right) + 2(2 + \xi) + \frac{(3 + \xi)(-m^2 + p^2) \ln\left(\frac{m^2}{m^2 - p^2}\right)}{p^2} \right]
\end{aligned}$$

Exercise: The electron self-energy function is gauge dependent. But the position of the pole in the complex p^2 plane of the full propagator $S(\boldsymbol{p})^{-1} = \boldsymbol{p} - m - \Sigma(\boldsymbol{p})$ should be gauge-independent. Compute the one-loop correction to the position of the pole and show that the result is gauge-independent.

5.4 Electron anomalous magnetic moment (**Projector**)

In this example, we will compute the QED contribution to the electron anomalous magnetic moment, which is obtained from the $q^2 \rightarrow 0$ limit of the Pauli form factor $F_2(q^2)$. We will perform this calculation by projecting out the appropriate form factor using **Projector**.




Figure 4: One-loop vacuum polarization in QED

Set up the problem by applying the QED Feynman rules to write down the integral we need to do:

$$\begin{aligned}
 & \bar{u}(\mathbf{p}') ie\Gamma^\mu u(\mathbf{p}) \\
 &= \mu^{2\epsilon} \int \frac{d^d k}{(2\pi)^d} \bar{u}(\mathbf{p}') \left[ie\gamma^\rho \frac{i(\not{p}' - \not{k} + m)}{(p' - k)^2 - m^2} ie\gamma^\mu \frac{i(\not{p} - \not{k} + m)}{(p - k)^2 - m^2} ie\gamma^\nu \right] u(\mathbf{p}) \frac{-ig_{\rho\nu}}{k^2} \\
 &= e^3 \mu^{2\epsilon} \int \frac{d^d k}{(2\pi)^d} \bar{u}(\mathbf{p}') \left[\frac{\gamma^\rho (\not{p}' - \not{k} + m) \gamma^\mu (\not{p} - \not{k} + m) \gamma^\nu g_{\rho\nu}}{[(p' - k)^2 - m^2][(p - k)^2 - m^2]k^2} \right] u(\mathbf{p}) \quad (12)
 \end{aligned}$$

By Lorentz covariance and gauge invariance, we expect the vertex function Γ^μ to take the following form:

$$\bar{u}(\mathbf{p}') \Gamma^\mu u(\mathbf{p}) = \bar{u}(\mathbf{p}') \left[F_1(q^2) \gamma^\mu + F_2(q^2) \frac{i}{2m} \sigma^{\mu\nu} q_\nu \right] u(\mathbf{p})$$

To project out $F_2(q^2)$ from the vertex function, we replace the external spinors u and \bar{u} with the appropriate projector $\mathcal{F}_\mu^{[F_2]}(p, p')$, and take the trace:

$$F_2(q^2) = \text{Tr}[\Gamma^\mu \mathcal{F}_\mu^{[F_2]}(p, p')]$$

Applying this to both sides of (12), we have

$$ieF_2(q^2) = e^3 \mu^{2\epsilon} \int \frac{d^d k}{(2\pi)^d} \frac{\text{Tr}[\gamma^\rho (\not{p}' - \not{k} + m) \gamma^\mu (\not{p} - \not{k} + m) \gamma^\nu \mathcal{F}_\mu^{[F_2]}(p, p')] g_{\rho\nu}}{[(p' - k)^2 - m^2][(p - k)^2 - m^2]k^2}. \quad (13)$$

Now we can use *Package-X* to compute this integral. We will start with the numerator, with $p = \mathbf{p1}$ and $p' = \mathbf{p2}$. Although unnecessary at this stage, I have also applied the on shell conditions and simplified the result. Notice, however, that I have not yet made the replacement $q^2 = 0$ for reasons that will soon become clear:

```
In[1]:= numerator = Contract[Spur[γρ, γ.(p2 - k) + m1, γμ, γ.(p1 - k) + m1, γν,
    Projector["F2", μ][{p1, m}, {p2, m}]] gρ,ν] /.
    {p1.p1 → m2, p2.p2 → m2, p1.p2 → -q.q/2 + m2} // Simplify
```

```
Out[1]= 
$$\frac{1}{q \cdot q (-4 m^2 + q \cdot q)^2} 8 m^2 (k \cdot k q \cdot q (-4 m^2 + q \cdot q) + k \cdot p2 q \cdot q (-4 m^2 + q \cdot q) +$$


$$(k \cdot p1)^2 (4 m^2 + (-2 + d) q \cdot q) + (k \cdot p2)^2 (4 m^2 + (-2 + d) q \cdot q) +$$


$$k \cdot p1 (q \cdot q (-4 m^2 + q \cdot q) + k \cdot p2 (-8 m^2 + 2 d q \cdot q)))$$

```

Consult the documentation page for **Projector** for a complete listing of available projectors.

Note how the function **Projector**["F2", μ][{p1, m}, {p2, m}] is used in the input. The first argument is a `String` and identifies the projector, and the second argument indicates the vector index. The kinematic information about the external spinors are given in the final arguments; the first momentum and mass always correspond to the unbarred spinor $u(\mathbf{p})$, and the second momentum and mass always correspond to the barred spinor $\bar{u}(\mathbf{p}')$.

Before we proceed, I want to alert you to a potential complication afflicting integrals involving projections. We would eventually like to take the $q^2 \rightarrow 0$ limit to obtain the anomalous magnetic moment. But the leading factor $\frac{1}{q^2(-4m^2+q^2)^2}$ in our **numerator** suggests that the integral will be singular at $q^2 = 0$. This singularity is fake (removable): its appearance is an artifact of making a projection, and should disappear upon integrating over the loop momentum k .

I have programmed **LoopIntegrate** to automatically decompose the integral differently than usual when there are kinematic singularities in the numerator so that they may readily be removed algebraically from the output (see the appendix below). With this in mind, we can proceed with computing the integral. Insert **numerator** into **LoopIntegrate**, and replace the scalar products using the on shell conditions:

```
In[2]:= LoopIntegrate[numerator, k, {{p2 - k, m}, {p1 - k, m}, {k, 0}}] /.
    {p1.p1 → m2, p2.p2 → m2, p1.p2 → -q.q/2 + m2} // Simplify
```

```
Out[2]= 
$$2 m^2 (2 \text{PVC}[0, 0, 1, m^2, q \cdot q, m^2, 0, m, m] +$$


$$(-2 + d) \text{PVC}[0, 0, 2, m^2, q \cdot q, m^2, 0, m, m] + 2 \text{PVC}[0, 0, 2, m^2, q \cdot q, m^2, 0, m, m] -$$


$$4 \text{PVC}[0, 1, 1, m^2, q \cdot q, m^2, 0, m, m] + 2 d \text{PVC}[0, 1, 1, m^2, q \cdot q, m^2, 0, m, m] -$$


$$2 \text{PVC}[0, 2, 0, m^2, q \cdot q, m^2, 0, m, m] + d \text{PVC}[0, 0, 2, m^2, q \cdot q, m^2, 0, m, m])$$

```

The final command **Simplify** successfully removes the singularity at $q^2 = 0$. In my experience, *Mathematica*'s algebraic functions **Cancel**, **Factor**, and **FactorSquareFree** all remove the singularity from the output, and may provide faster alternatives.

The replacement $q \cdot q \rightarrow 0$ can now be made immediately.

```
In[4]:= % /. q.q → 0 // LoopRefine
```

```
Out[4]= 2
```

Remembering that there is an implicit $\frac{i}{16\pi^2}$, eqn (13) becomes

$$\begin{aligned} ieF_2(0) &= e^3 \frac{i}{16\pi^2} \left[-\text{computer output} \right] \\ &= e^3 \frac{i}{16\pi^2} \times 2, \\ \text{or, } F_2(0) &= \frac{e^2}{8\pi^2} = \frac{\alpha_{\text{em}}}{2\pi}. \end{aligned} \tag{14}$$

Exercise: Return to the previous example (Electron self-energy function), and use **Projector** to separately extract the coefficients of p and $1m$ from $i\Sigma(p)$.

Appendix: Notes on the internals of **LoopIntegrate**

Usually, **LoopIntegrate** tries to cancel scalar products involving the integration variable k common to the numerator and the denominator, controlled by the option **Cancel**, before expressing the result in terms of Passarino-Veltman functions. For example,

$$\begin{aligned} \mu^{2\epsilon} \int \frac{d^d k}{(2\pi)^d} \frac{k^2}{[k^2 - m^2][(k+p)^2 - m^2]} \\ &= \mu^{2\epsilon} \int \frac{d^d k}{(2\pi)^d} \frac{(k^2 - m^2) + m^2}{[k^2 - m^2][(k+p)^2 - m^2]} \\ &= \mu^{2\epsilon} \int \frac{d^d k}{(2\pi)^d} \frac{1}{(k+p)^2 - m^2} + \mu^{2\epsilon} \int \frac{d^d k}{(2\pi)^d} \frac{m^2}{[k^2 - m^2][(k+p)^2 - m^2]} \\ &\longrightarrow A_0(m) + m^2 B_0(p^2; m, m). \end{aligned}$$

Under most circumstances, this leads to substantially increased performance and quality of the output of **LoopRefine**. However this behavior causes removable kinematic singularities generated by projectors to persist. Therefore, with the default setting **Cancel** → **Automatic**, if the numerator contains kinematic singularities, **LoopIntegrate** will not cancel these scalar products so that it is easier to remove these kinematic singularities.

Exercise: Investigate what happens when the option **Cancel** → **True** is set to **LoopIntegrate** when trying to make a projection. Observe that it is not possible to make the replacement **q.q** → 0 without generating **Power::infy** errors. Try using **LoopRefineSeries** on the output of **LoopIntegrate** to obtain the leading $q^2 \rightarrow 0$ behavior of the integral (feel free to consult the help page for **LoopRefineSeries**).

5.5 One-loop QCD correction to $\gamma^* \rightarrow q\bar{q}$ (**FermionLine**, u, v)

In this section, we will compute the one-loop QCD correction to the QED vertex function $\gamma^* \rightarrow q\bar{q}$. This calculation is the virtual component of the fully inclusive cross section for the process $e^+e^- \rightarrow \text{hadrons}$ at $O(\alpha_s)$ shown below.




Figure 5: Virtual QCD correction to $\gamma^* \rightarrow q\bar{q}$

We will approximate the quarks (electric charge Q_q) as being massless, and compute the vertex function in the Feynman gauge:

$$\begin{aligned}
 & -iQ_q e \bar{u}(\mathbf{p}') \Gamma^\mu u(\mathbf{p}) \\
 &= \mu^{2\epsilon} \int \frac{d^d k}{(2\pi)^d} \bar{u}(\mathbf{p}) \left[(-ig\gamma^\nu T^a) \frac{i(\mathbf{p} + \mathbf{k})}{(p+k)^2} (-iQ_q e \gamma^\mu) \frac{i(-\mathbf{p}' + \mathbf{k})}{(-p'+k)^2} (-ig\gamma^\rho T^b) \right] v(\mathbf{p}') \frac{-i\delta^{ab} g_{\nu\rho}}{k^2} \\
 &= -g^2 C_F Q_q e \mu^{2\epsilon} \int \frac{d^d k}{(2\pi)^d} \frac{\bar{u}(\mathbf{p}) \gamma^\nu (\mathbf{p} + \mathbf{k}) \gamma^\mu (-\mathbf{p}' + \mathbf{k}) \gamma^\rho v(\mathbf{p}') g_{\nu\rho}}{(p+k)^2 (-p'+k)^2 k^2} \quad (15)
 \end{aligned}$$

In the final line, the color factors have been collected $T^a T^b \delta^{ab} = C_F$, and the integral is put in a form that is ready for input.

In *Package-X*, the product of Dirac matrices sandwiched between on shell spinors is entered using the **FermionLine** object like this (letting $p = \mathbf{p1}$ and $p' = \mathbf{p2}$):

```

In[1]:= num = <u[p1, 0],  $\gamma_\nu$ ,  $\gamma \cdot (\mathbf{p1} + \mathbf{k})$ ,  $\gamma_\mu$ ,  $\gamma \cdot (-\mathbf{p2} + \mathbf{k})$ ,  $\gamma_\rho$ , v[p2, 0]> g $\nu\rho$  // Contract

Out[1]= <u[p1, 0],  $\gamma_\rho$ ,  $\gamma \cdot k + \gamma \cdot \mathbf{p1}$ ,  $\gamma_\mu$ ,  $\gamma \cdot k - \gamma \cdot \mathbf{p2}$ ,  $\gamma_\rho$ , v[p2, 0]>

```

Additional input aliases listed in Table 3 have been defined for rapid notebook entry of the on shell spinors at the ends of the string of Dirac matrices. Notice that $u[\mathbf{p}, \mathbf{m}]$ and $v[\mathbf{p}, \mathbf{m}]$ take two mandatory arguments which specify the on shell momentum and mass. The syntax for the intervening Dirac matrices are the same as those for **Spur** and **DiracMatrix**. *Package-X* will automatically parse this input as a **FermionLine** object.

Spinor	Alias	StandardForm
$\bar{u}(\mathbf{p})$	<code>[esc] ub [esc]</code>	$\langle u[p, m]$
$\bar{v}(\mathbf{p})$	<code>[esc] vb [esc]</code>	$\langle v[p, m]$
$u(\mathbf{p})$	<code>[esc] u [esc]</code>	$u[p, m] \rangle$
$v(\mathbf{p})$	<code>[esc] v [esc]</code>	$v[p, m] \rangle$

Table 3: Inputting u and v spinors

As an aside, we can apply **FermionLineReduce** to expand out the product of Dirac matrices:

```
In[2]:= num // FermionLineReduce
Out[2]= <u[p1, 0], γμ, v[p2, 0]>
        (-2 k.k + d k.k + 4 k.p1 - 4 k.p2 - 4 p1.p2) +
        (4 - 2d) <u[p1, 0], γ.k, v[p2, 0]> kμ -
        4 <u[p1, 0], γ.k, v[p2, 0]> p1μ +
        4 <u[p1, 0], γ.k, v[p2, 0]> p2ν
```

Consult the documentation page for further details on **FermionLineReduce**.

In addition to using the Dirac algebra, **FermionLineReduce** will apply the spinor equations of motion

$$\not{p} u(\mathbf{p}) = m u(\mathbf{p}), \quad \not{p} v(\mathbf{p}) = -m v(\mathbf{p}),$$

express products of Dirac matrices in canonical SVTAP form

$$\gamma^{\mu_1} \gamma^{\mu_2} \gamma^{\mu_3} \dots = A^{\mu_1 \mu_2 \mu_3 \dots} \mathbb{1} + (B^{\mu_1 \mu_2 \mu_3 \dots})_{\nu} \gamma^{\nu} + (C^{\mu_1 \mu_2 \mu_3 \dots})_{\nu \rho} \sigma^{\nu \rho} + (D^{\mu_1 \mu_2 \mu_3 \dots})_{\nu} \gamma^{\nu} \gamma_5 + E^{\mu_1 \mu_2 \mu_3 \dots} \gamma_5,$$

and if possible, apply the Gordon identities

$$\bar{u}(\mathbf{p}')(\not{p}' + \not{p})u(\mathbf{p}) = 2m \bar{u}(\mathbf{p}')\gamma^{\mu}u(\mathbf{p}) - i \bar{u}(\mathbf{p}')\sigma^{\mu\nu}(\not{p}' - \not{p})u(\mathbf{p}). \quad (16)$$

These manipulations are automatically carried out by **LoopIntegrate** when **FermionLine** appears in the numerator:

```
In[3]:= integral = LoopIntegrate[num, k, {{p1 + k, 0}, {-p2 + k, 0}, {k, 0}}] /.
        {p1.p1 -> 0, p2.p2 -> 0, p1.p2 -> q.q/2}
```

```
Out[3]= <u[p1, 0], γμ, v[p2, 0]> (4 PVB[0, 0, 0, 0, 0] - 6 PVB[0, 0, q.q, 0, 0] +
        d PVB[0, 0, q.q, 0, 0] - 2 q.q PVC[0, 0, 0, 0, q.q, 0, 0, 0, 0] +
        4 PVC[1, 0, 0, 0, q.q, 0, 0, 0, 0] - 2 d PVC[1, 0, 0, 0, q.q, 0, 0, 0, 0])
```

After applying the on shell conditions, call **LoopRe fine** to obtain an analytic expression for this integral:

```
In[3]:= LoopRe fine[integral]
```

```
Out[3]= <u[p1, 0], γμ, v[p2, 0]>
        \left( -8 + \frac{\pi^2}{6} - 3 \left( \frac{1}{\epsilon} + \text{Log} \left[ -\frac{\mu^2}{q.q} \right] \right) - 2 \left( \frac{1}{\epsilon^2} + \frac{\text{Log} \left[ -\frac{\mu^2}{q.q} \right]}{\epsilon} + \frac{1}{2} \text{Log} \left[ -\frac{\mu^2}{q.q} \right]^2 \right) \right)
```

Consult the documentation page for **LoopIntegrate** or **ε** for assistance in interpreting $1/\epsilon^2$ in the output.

From this output, we can write down the result for the virtual QCD correction to $\gamma^* \rightarrow \bar{q}q$:

$$\begin{aligned}
-iQ_q e \bar{u}(\mathbf{p}') \Gamma^\mu u(\mathbf{p}) = \bar{u}(\mathbf{p}') \gamma^\mu u(\mathbf{p}) \frac{ig^2 C_F Q_q e}{16\pi^2} \Big\{ & -8 + \frac{\pi^2}{6} - 3 \left(\frac{1}{\epsilon} - \gamma_E + \ln(4\pi) + \ln\left(\frac{\mu^2}{-q^2}\right) \right) \\
& - 2 \left[\frac{1}{\epsilon^2} + \frac{1}{\epsilon} \left(-\gamma_E + \ln(4\pi) \right) + \frac{\gamma_E^2}{2} - \gamma_E \ln(4\pi) + \frac{1}{2} \ln^2(4\pi) \right. \\
& \left. + \left(\frac{1}{\epsilon} - \gamma_E + \ln(4\pi) \right) \ln\left(\frac{\mu^2}{-q^2}\right) + \frac{1}{2} \ln^2\left(\frac{\mu^2}{-q^2}\right) \right] \Big\}
\end{aligned}$$

Exercise: The $1/\epsilon$ and $1/\epsilon^2$ poles in the output above arise from both UV and IR divergences. Set the option **Part** \rightarrow **UVDivergent** or **Part** \rightarrow **IRDivergent** to **LoopRefine** to compute these parts of the integral separately.

Exercise: Compute the vertex function retaining the final state quark masses in the general covariant ξ gauge. Separate the $1/\epsilon$ pole based on their UV or IR origin.

Exercise: Return to the previous example (Electron anomalous magnetic moment) and use the **FermionLine** object to compute the entire QED vertex function $\Gamma^\mu(q^2)$ for massive electrons. Notice that both form factors $F_1(q^2)$ and $F_2(q^2)$ are simultaneously obtained in this way.

5.6 Tree-level $Z^0 \rightarrow f\bar{f}$ decay width (**Discontinuity**)

In this section we will re-derive the tree-level $Z^0 \rightarrow f\bar{f}$ partial width (4) by calculating the imaginary part of a one-loop integral. The basic idea is that by S -matrix unitarity, the discontinuity across the normal threshold cut (twice the imaginary part) of a one loop integral gives the square of the lower order amplitude, with the phase space integrations already done. Pictorially, this corresponds to cutting the internal lines of a Feynman diagram:

$$\mu \xrightarrow{q} \text{loop} \xrightarrow{\nu} = \int \frac{d^3\mathbf{k}}{(2\pi)^3} \frac{1}{2E_{\mathbf{k}}} \left(\mu \xrightarrow{q} \text{tree} \right) \left(\nu \xrightarrow{q} \text{tree} \right)^*$$

After dividing the discontinuity by the flux factor $2m_Z$ for the decay process, the partial width is obtained:

$$\Gamma = \frac{-i}{\text{Flux}} \text{Disc } \mathcal{M}. \quad (17)$$

Using the Feynman rules for electroweak theory, the one-loop amplitude is

$$i\mathcal{M} = (-)\mu^{2\epsilon} \int \frac{d^d k}{(2\pi)^d} N_c \text{Tr} \left[\frac{-ie}{\sin^2(2\theta_W)} \gamma^\nu (g_V^f - g_A^f \gamma_5) \frac{i(\mathbf{k} + m_f)}{k^2 - m_f^2} \right. \\ \left. \times \frac{-ie}{\sin^2(2\theta_W)} \gamma^\mu (g_V^f - g_A^f \gamma_5) \frac{i(\mathbf{k} - \mathbf{q} + m_f)}{(k - q)^2 - m_f^2} \right] \epsilon_\nu^*(\mathbf{q}) \epsilon_\mu(\mathbf{q}),$$

or after pulling out the overall constants,

$$i\mathcal{M} = -\frac{e^2 N_c}{\sin^2(2\theta_W)} \epsilon_\nu^*(\mathbf{q}) \epsilon_\mu(\mathbf{q}) \\ \mu^{2\epsilon} \int \frac{d^d k}{(2\pi)^d} \frac{\text{Tr}[\gamma^\nu (g_V^f - g_A^f \gamma_5)(\mathbf{k} + m_f) \gamma^\mu (g_V^f - g_A^f \gamma_5)(\mathbf{k} - \mathbf{q} + m_f)]}{(k^2 - m_f^2)((k - q)^2 - m_f^2)} \quad (18)$$

Input the one-loop integral, setting aside the prefactor in the first line, and apply the on shell condition $q^2 = m_Z^2$:

```

In[1]:= LoopIntegrate[Spur[γν, gV1 - gAγ5, k.γ + mf1, γμ, gV1 - gAγ5, (k - q).γ + mf1],
  k, {{k, mf}, {k - q, mf}}] /. q.q -> mZ^2

Out[1]= qμqν ( (8 gA^2 + 8 gV^2) PVB[0, 1, mZ^2, mf, mf] + (8 gA^2 + 8 gV^2) PVB[0, 2, mZ^2, mf, mf] ) +
  gμ,ν ( (-4 gA^2 - 4 gV^2) PVA[0, mf] + (-8 gA^2 mf^2 + 2 gA^2 mZ^2 + 2 gV^2 mZ^2)
  PVB[0, 0, mZ^2, mf, mf] + (8 gA^2 + 8 gV^2) PVB[1, 0, mZ^2, mf, mf] )

```

Now, we would like to obtain the discontinuity across the normal threshold cut of this expression. Apply **LoopRe fine** to the expression, but to limit the output to just the discontinuity set the option **Part** \rightarrow **Discontinuity**[mZ²]:

```
In[2]:= disc = LoopRe fine [% , Part → Discontinuity [mZ2] , Organization → Function]
```

```
Out[2]= 2 i π HeavisideTheta [-4 m f2 + m Z2]
        ( - 4 (g A2 + g V2) √{m Z2 (-4 m f2 + m Z2) (2 m f2 + m Z2) qμ qν +
          3 m Z4 +
          4 √{m Z2 (-4 m f2 + m Z2) (-4 g A2 m f2 + 2 g V2 m f2 + g A2 m Z2 + g V2 m Z2) gμ, ν )
          3 m Z2 )
```

Notice that the argument of **Discontinuity** indicates the channel variable across which the discontinuity should be taken. In this example, a self-energy type integral, there is only one channel $q^2 = m_Z^2$. But problems involving more external lines feature several channel variables.

This result is essentially the square of the tree-level matrix element with the phase space integrals already completed. Notice that the **HeavisideTheta** function has support only when the channel is open, $m_Z^2 \geq 4m_f^2$.

The final step in obtaining the partial width is to perform the polarization average of the initial Z^0 boson. Contract the result with the completeness relation $(-g_{\mu\nu} + k_\mu k_\nu / m_Z^2)$, and divide by 3 to account for the averaging. Finally, make the replacement $q^2 = m_Z^2$ and apply **LoopRe fine** to take the limit $d \rightarrow 4$ and also to organize the result:

```
In[2]:= Contract [disc * 1/3 (-gμ, ν + qμ qν / mZ2) ] /. q . q → mZ2 // LoopRe fine
```

```
Out[2]= - 1 / (3 m Z2) 8 i √{m Z2 (-4 m f2 + m Z2)
        (- 4 g A2 m f2 + 2 g V2 m f2 + g A2 m Z2 + g V2 m Z2) π HeavisideTheta [-4 m f2 + m Z2]
```

It is now a matter of restoring the previously omitted prefactors from (18), (17) and the implicit normalization $\frac{1}{16\pi^2}$ (the i in $\frac{i}{16\pi^2}$ cancels with that in front of $i\mathcal{M}$) to arrive at the final result,

$$\begin{aligned} \Gamma &= -\frac{e^2 N_c}{\sin^2(2\theta_W)} \frac{1}{16\pi^2} \frac{-i}{2m_Z} \left[-\text{computer output} \right] \\ &= \frac{\alpha m_Z}{\sin^2(2\theta_W)} \left(\frac{N_c}{3} \right) \sqrt{1 - \frac{4m_f^2}{m_Z^2}} \left[((g_V^f)^2 + (g_A^f)^2) + 2((g_V^f)^2 - 2(g_A^f)^2) \frac{m_f^2}{m_Z^2} \right], \end{aligned}$$

reproducing eq. (4) at the beginning of this primer.

For another, more involved example of an application of **Discontinuity**, see the tutorial “[Tree-Level \$e^+e^- \rightarrow \gamma\gamma\$ Cross Section by The Optical Theorem](#)” near the bottom of the guide page.

6 Tips for a pleasant computing experience

The documentation pages available within the *Wolfram Documentation Center* provide a lot of details on each symbol defined by *Package-X*. Please browse through them to familiarize yourself with the available bells and whistles since they may come in handy. In this section, I've gathered supplementary tips and pointers to make your computing experience as pleasant as possible.

Refrain from using special formatting for symbols (like \mathbf{m}_H and \mathbf{p}')

This is a tip not only for *Package-X* users, but also for *Mathematica* users in general. Beginners will try to use *Mathematica*'s formatting capabilities to write code using subscripted variables like \mathbf{m}_H or primed variables like \mathbf{p}' . These are not interpreted by *Mathematica* as symbols, but rather as expressions:

n00b input	internal rep.	better input	internal rep.
\mathbf{m}_H	LTensor[m,H] in <i>Package-X</i>	\mathbf{mH}	mH
\mathbf{p}'	Derivative[1][p]	\mathbf{p}'	p'

This can cause unexpected behavior, for example, in *Package-X*

```
In[1]:=  $\mathbf{m}_H^2$  // Contract
Out[1]= m.m
```

Get in the habit of using unformatted symbols for your variables, opting instead for \mathbf{mH} and \mathbf{p}' (alias: $\mathbf{p}' = \text{esc} \text{ ' } \text{esc}$).

Reuse and store expressions generated by **LoopRe fine**

When working on extensions of the Standard Model, it could happen that a basic integral can be used to determine contributions from several different particle species. In this case, you can store the general formula as a function of multiple parameters, and then reuse it for the various particles in the loop.

For example, you can obtain the general formula for the transverse part of the *W*-boson vacuum polarization function at zero external momentum $\Pi_T^{WW}(0)$ as a function of internal fermion masses m_1 and m_2 :

```
In[1]:= int =
Transverse [ LoopIntegrate [ Spur [  $\frac{-i g}{\sqrt{2}} \gamma_\mu$ , PL,
 $i (k \cdot \gamma + m_1 1)$ ,  $\frac{-i g}{\sqrt{2}} \gamma_\nu$ , PL,  $i ((k + p) \cdot \gamma + m_2 1)$  ], k,
{ {k, m1}, {k + p, m2} } ] ] / . p.p -> 0;
```

Then store the output of **LoopRe fine** as $\mathbf{f}[m_1, m_2]$:

```
In[2]:= f[m1_, m2_] = LoopRe fine[int]

Out[2]=  $-\frac{1}{4} g^2 (m_1^2 + m_2^2) + \frac{g^2 m_1^4 \text{Log}\left[\frac{m_1^2}{m_2^2}\right]}{2 (m_1^2 - m_2^2)} -$ 
 $\frac{1}{2} g^2 (m_1^2 + m_2^2) \left( \frac{1}{\epsilon} + \text{Log}\left[\frac{\mu^2}{m_2^2}\right] \right)$ 
```

Notice the use of **Set** (=) instead of **SetDelayed** (:=) so that the right hand side is evaluated before storing the expression into **f**. You may add a semicolon (;) at the end of the line to suppress the output.

If you have a massless particle in the loop (say, $m_2 = 0$), this formula will lead to a $\ln(0)$ error. It's easy enough to define a special case for **f**:

```
In[3]:= f[m1_, 0] = LoopReFine[int /. m2 -> 0]
```

$$\text{Out[3]} = -\frac{1}{4} g^2 m_1^2 - \frac{1}{2} g^2 m_1^2 \left(\frac{1}{\epsilon} + \text{Log} \left[\frac{\mu^2}{m_1^2} \right] \right)$$

(*Exercise*: How would you define a special case when $m_1 = m_2$?) Then you can add up the fermionic contributions to $\Pi_T^{WW}(0)$, assuming massless neutrinos, without making multiple calls to **LoopReFine**:

```
In[4]:= 3 f[mt, mb] + 3 f[mc, ms] + 3 f[mu, md] +
        f[mτ, 0] + f[mμ, 0] + f[me, 0]
```

Apply **LoopReFine** to special sets of Feynman integrals

In gauge theories, several different topologies of loop integrals usually contribute to a single process (or Green's function), which combine neatly after several cancellations. If you expect such simplifications or cancellations, it may be helpful to apply **LoopReFine** to the sum of **LoopIntegrate** output of the individual integrals.

Pay attention when using γ_5

In *Package-X*, γ_5 is defined to anticommute with all Dirac matrices in $d = 4 - 2\epsilon$ dimensions. You should be aware that this implementation of γ_5 may lead to algebraic inconsistencies starting at $\mathcal{O}(\epsilon)$, which means that finite parts of integrals exhibiting $1/\epsilon$ poles might be incorrect. Here are some examples and suggested solutions to resolve this inconsistency within *Package-X*:

- Triangle integrals involving a closed fermion loop with two or three chiral vector vertices are susceptible to inconsistencies.

Recommended solution: Enforce the necessary Ward identities using Adler's method. See the tutorial "[Ward Identities and \$\gamma_5\$ in Dimensional Regularization](#)" listed near the bottom of the guide page in the documentation center for more information.

- Box integrals involving fermions and internal gauge bosons in the unitary gauge contain UV divergent $1/\epsilon$ poles at intermediate stages. Regardless of whether these poles cancel in the end, finite parts of these integrals are also susceptible to these inconsistencies.

Recommended solution: Perform the computation in a renormalizable gauge such as 't Hooft-Feynman gauge. Note that Landau gauge reintroduces the problem due to mass singular $1/\epsilon$ poles.

Although normally UV divergent, it is unusual for tadpole and bubble integrals to exhibit these inconsistencies due to the lower number of external invariants upon which they depend. For maximum control, you can set the global variable:

```
In[4]:= $DiracAlgebra = False;
```

This turns off all Dirac algebraic routines. Then the output of **LoopIntegrate** will be given in terms of unevaluated products of gamma matrices. It will be more verbose, but you can now manipulate the gamma matrices by hand.


Try different evaluation strategies to resolve numerical instabilities

It is quite common to encounter loop integrals with widely separated scales. In these cases, machine precision evaluation usually leads to numerical instabilities. Consider for example this expression obtained after applying **LoopRefine** to an (unspecified) integral:

```
In[1]:= f[s_, t_, m_] = LoopRefine[example, ExplicitC0 → None]
Out[1]= -  $\frac{s \text{DiscB}[s, m, m]}{s - t}$  -  $\frac{s \text{DiscB}[s, m, m]}{s - t}$  -  $2 m^2 \text{ScalarC0}[0, s, t, m, m, m]$ 
```

For very large **m**, the default machine precision evaluation exhibits numerical instabilities:


```
In[2]:= LogLinearPlot[Re[f[21.5, 1.5, m]], {m, 5, 108},
PlotRange → {2.5, 3.5}]
```



See the *Mathematica tutorial Arbitrary-Precision Numbers* for details.

Since all of *Package-X*'s special functions support *Mathematica*'s arbitrary-precision arithmetic, you can use it (at the expense of computation speed) to achieve numerical stability. Here we supply numbers with a precision level of 20:

```
In[3]:= LogLinearPlot[Re[f[21.5`20, 1.5`20, m]], {m, 5, 108},
PlotRange → {2.5, 3.5}, WorkingPrecision → 20]
```



Alternatively, you could obtain an analytic approximation to the integral by using **LoopRefineSeries**:

```
In[4]:= fApprox[s_, t_, m_] =
LoopRefineSeries[example, {s, 0, 1}, {t, 0, 1}] // Normal
Out[4]= 3 -  $\frac{t}{12 m^2} + s \left( -\frac{t}{12 m^2} - \frac{t}{180 m^4} \right)$ 
```

A plot of this expression using machine precision yields a smooth curve visually identical to the one obtained by using arbitrary-precision above.

Link *Package-X* to third party numerical libraries

Package-X is optimized to give analytic expressions when a compact one is expected. For NLO cross sections with full kinematic dependence, it is known that the formulae are usually not compact. If you insist on using *Package-X* to compute these complicated quantities, I suggest linking *Package-X* to numerical libraries.

- **LOOPTOOLS** (<http://www.feynarts.de/looptools/>) by T. Hahn provides the most readily accessible numerical library that can be called within *Mathematica*, but is limited to the lower rank Passarino-Veltman functions that occur in renormalizable gauge theories. See the tutorial “[Linking Package-X to LoopTools](#)” listed near the bottom of the guide page in the documentation center for more information.
- **COLLIER** (<https://collier.hepforge.org>) by A. Denner, S. Dittmaier, and L. Hofer is a more recent library that is capable of numerically evaluating any Passarino-Veltman coefficient function and features numerous algorithms to avoid numerical instabilities. **COLLIERLINK** is a recently released expansion pack for *Package-X* which provides an interface to the **COLLIER** library. See the tutorial “[Linking Package-X to COLLIER](#)” for more information.

In both cases, using these library in conjunction with *Package-X* will allow you to directly evaluate the Passarino-Veltman functions **PVA**, **PVB**, **PVC**, **PVD**, without needing to call **LoopRefine**.

Cross check results within *Package-X*

It is usually possible to obtain equivalent results in several different ways using *Package-X*. This redundancy can be used to cross check your calculation, and to isolate any internal inconsistencies/bugs within *Package-X*. I mention a few here that you may find useful.

For cross-checking analytical integrations, expressions generated by **LoopIntegrate** by toggling the option **Cancel** to **True** and **False** will contain different combinations of Passarino-Veltman functions. Furthermore, if your integrals contain linearly dependent denominator factors (*e.g.* with one or more vanishing external momenta), toggling the option **Apart** to **LoopIntegrate** will also generate different combinations of Passarino-Veltman functions. In all cases, the result of applying **LoopRefine** should *always* generate equivalent results, and can be used for checking the internal consistency of *Package-X*.

For numerical cross-checks of special functions like **ScalarC0**, **ScalarD0**, *etc.*, it is helpful to know that the numerical implementation and the analytic expression generated by **C0Expand** and **D0Expand** are of different origin. Evaluating the analytic representations numerically should return the same result as evaluating the special functions directly.

When projecting form factors with **Projector**, you should be able to read off the same form factors by performing the integral with a **FermionLine** object in the numerator.

A Space-time conventions

For reference, the conventions for space-time quantities are summarized in Table 4 below.

Quantity	Convention
Metric signature	$g_{\mu\nu} = \text{diag}(+, -, -, -)$
Spacetime dimension	$d = 4 - 2\epsilon$
Dirac matrix commutator	$\sigma_{\mu\nu} = \frac{i}{2}[\gamma_\mu, \gamma_\nu]$
Fifth gamma matrix	$\gamma_5 = i\gamma^0\gamma^1\gamma^2\gamma^3$
Chiral projectors	$\hat{P}_L = \frac{1}{2}(1 - \gamma_5), \hat{P}_R = \frac{1}{2}(1 + \gamma_5)$
Levi-civita symbol	$\epsilon^{0123} = +1$

Table 4: Conventions for space-time quantities